
Chapter 6 

Design_Organization

· Subprograms 

· Packages 

· Parametrization 

· Top level configuration 

· Design libraries 

Functions and Procedures

a_gt_b = a. gt + b'. gt+ a.b' 

a_eq_b = a.b.eq + a'.b'.eq 

a_lt_b = b.lt + a'.lt + b.a' 

ARCHITECTURE functional OF bit_comparator IS 

FUNCTION fgl (w, x, gl : BIT) RETURN BIT IS 

BEGIN 

RETURN (w AND gl) OR (NOT x AND gl) OR (w AND NOT x); 

END fgl; 

FUNCTION feq (w, x, eq : BIT) RETURN BIT IS

BEGIN
RETURN (w AND x AND eq) OR (NOT w AND NOT x AND eq);

END feq; 

BEGIN
a_gt_b <= fgl (a, b, gt) AFTER 12 NS;

a_eq_b <= feq (a, b, eq) AFTER 12 NS;

a_lt_b <= fgl (b, a, lt) AFTER 12 NS;

END functional; 

Syntax 

· Bodies of Subprograms (Functions & Procedures) are Sequential (Statements are Executed Sequentially)

· Subprograms (Functions & Procedures) are Computed in ZERO Real Time (Simulation time is the same B4 and after Function Computation).

· Delays Here (Functional Architecture) are not as Accurate as the Structural (gate_level Archuitecture) Model (0 to 1 Delay different from Delay 1 to 0, etc.)

· Assignments to Output Signals call Functions

Nibble_Comparator Using functional Architecture of Bit_Comparator
ARCHITECTURE structural OF nibble_comparator IS 

Component comp1

PORT (a, b, gt, eq, lt : IN BIT; a_gt_b, a_eq_b, a_lt_b : OUT BIT);

END Component; 

FOR ALL : comp1 USE ENTITY work.bit_comparator (functional);

CONSTANT n : INTEGER := 4;

SIGNAL im : BIT_VECTOR ( 0 TO (n-1)*3-1);

BEGIN
c_all: FOR i IN 0 TO n-1 GENERATE

l: IF i = 0 GENERATE
least: comp1 PORT MAP (a(i), b(i), gt, eq, lt, im(0), im(1), im(2) );

END GENERATE;

m: IF i = n-1 GENERATE
most: comp1 PORT MAP
(a(i), b(i), im(i*3-3), im(i*3-2), im(i*3-1), a_gt_b, a_eq_b, a_lt_b);

END GENERATE;

r: IF i > 0 AND i < n-1 GENERATE
rest: comp1 PORT MAP
(a(i), b(i), im(i*3-3), im(i*3-2), im(i*3-1), im(i*3+0), im(i*3+1), 

 im(i*3+2) );

END GENERATE;

END GENERATE;

END structural; 



· A nibble comparator is based on the functional ARCHITECTURE 

· Only bindings need to be changed (Configuration Statement) 

TEST BENCH SubPrograms
ARCHITECTURE procedural OF nibble_comparator_test_bench is 

TYPE integers IS ARRAY (0 TO 12) OF INTEGER; 

PROCEDURE apply_data (SIGNAL target : OUT BIT_VECTOR   (3 DOWNTO 0); CONSTANT values : IN integers; 

                               CONSTANT period : IN TIME) IS 

VARIABLE j : INTEGER; VARIABLE tmp : INTEGER := 0; 

VARIABLE buf : BIT_VECTOR (3 DOWNTO 0); 

BEGIN
FOR i IN 0 TO 12 LOOP
            tmp := values(i);                  j := 0;

            WHILE j <= 3 LOOP
-- Only 4-Bit Numbers
                     IF (tmp MOD 2 = 1) THEN buf (j) := '1'; 

                     ELSE buf (j) := '0'; END IF;

                     tmp := tmp / 2;               j := j + 1;

            END LOOP;

            target <= TRANSPORT buf AFTER i * period;

           END LOOP;

END apply_data; 

Component comp4 PORT (a, b : IN bit_vector (3 DOWNTO 0);

gt, eq, lt : IN BIT;   a_gt_b, a_eq_b, a_lt_b : OUT  BIT);
END Component; 

FOR a1 : comp4 USE ENTITY work.nibble_comparator(structural); 

SIGNAL a, b : BIT_VECTOR (3 DOWNTO 0);

SIGNAL eql, lss, gtr : BIT;

SIGNAL vdd : BIT := '1';                          SIGNAL gnd : BIT := '0'; 

BEGIN
a1: comp4 PORT MAP (a, b, gnd, vdd, gnd, gtr, eql, lss); 

apply_data (a, 0&15&15&14&14&14&14&10&00&15&00&00&15, 500 NS); 

apply_data (b, 0&14&14&15&15&12&12&12&15&15&15&00&00, 500 NS);

END procedural; 

TEST BENCH SubPrograms
BEGIN
a1: comp4 PORT MAP (a, b, gnd, vdd, gnd, gtr, eql, lss); 

apply_data (a, 0&15&15&14&14&14&14&10&00&15&00&00&15, 500 NS); 

apply_data (b, 0&14&14&15&15&12&12&12&15&15&15&00&00, 500 NS);

END procedural; 

· Two concurrent procedure calls of apply_data. 

· Procedure uses integers  and places binary equivalent on target signal

· Invoked at initialization or when inputs change (Here Called Once)

· Use of Function simplifies bit_comparator description 

· Price paid is Reduced Delay Accuracy in simulation 

Useful Utilities
PROCEDURE bin2int (bin : in BIT_VECTOR; int : out INTEGER) IS
VARIABLE result: INTEGER;

BEGIN
result := 0;

FOR i IN bin'Range LOOP 
IF bin(i) = '1' THEN result := result + 2**i; END IF;

END LOOP;

int := result;

END bin2int; 


Function Bin2Int (bin: BIT_VECTOR) Return INTEGER is

Variable Vec: BIT_VECTOR(bin'Length-1 DownTo 0); 


Variable SUM: INTEGER := 0; 

Begin 



Vec := bin; 



For i in Vec'Length-1 DownTo 0 LOOP 




SUM :=2*SUM;




IF Vec(i) = '1' then SUM := SUM + 1; 




end if; 



END LOOP; 



return SUM; 

END Bin2Int;

· Function Bin2Int is Better Written than Procedure Bin2Int (Why?) ( It is independent of the Actual Array Size and Index Range as well as More Computationally Efficient.

· Procedure Bin2Int Can work for any Array Size, but the LSB of the input Vector is Bit 0.  Thus, it won’t accept a vector of range (1 To 8) or (8 downto 1) ( Must be converted to (0 To 7) or (7 DownTo 0) before calling the procedure.

Function Bin2Int (bin: BIT_VECTOR) Return INTEGER is

Variable Result: INTEGER := 0; 


Variable SUM: INTEGER := 1; 

Begin 



For i in Bin’LOW To Bin’High LOOP 




IF Bin(i) = '1' then  Result:= Result + SUM ; 




end if; 




SUM :=2*SUM;



END LOOP; 



return Result; 

END Bin2Int;

PROCEDURE apply_data (SIGNAL target : OUT BIT_VECTOR 

                             (3 DOWNTO 0); CONSTANT values : IN integers; 

                              CONSTANT period : IN TIME) IS
VARIABLE buf : BIT_VECTOR (3 DOWNTO 0);

BEGIN 

FOR i IN 0 TO 12 LOOP 

int2bin (values(i), buf); -- buf := int2bin(values(i));

target <= TRANSPORT buf AFTER i * period;

END LOOP;

END apply_data;

The Following Procedure is Size Independent but not Range Independent

PROCEDURE int2bin (int : IN INTEGER; bin : OUT BIT_VECTOR) IS 

VARIABLE tmp : INTEGER;

VARIABLE buf : BIT_VECTOR (bin'RANGE);

BEGIN
tmp := int;

FOR i IN 0 TO (bin'LENGTH - 1) LOOP
IF (tmp MOD 2 = 1) THEN bin (i) := '1'; ELSE bin (i) := '0'; END IF;

tmp := tmp / 2;

END LOOP;

END int2bin; 

The Following Procedure is Both Size and Range Independent 

PROCEDURE int2bin (int : IN INTEGER; bin : OUT BIT_VECTOR) IS 

VARIABLE tmp : INTEGER;

BEGIN
tmp := int ;

FOR i IN bin'LOW TO bin'HIGH LOOP
IF (tmp MOD 2 = 1) THEN bin (i) := '1'; ELSE bin (i) := '0'; END IF;

tmp := tmp / 2;

END LOOP;

END int2bin; 

Design Packages

PACKAGE simple_gates IS 
Component n1 PORT (i1: IN BIT; o1: OUT BIT); end Component;
 
Component n2 PORT (i1, i2: IN BIT; o1: OUT BIT); end Component;
 
Component n3 PORT (i1, i2, i3: IN BIT; o1: OUT BIT); end


 Component; 












END simple_gates; 

USE WORK.simple_gates.ALL;







 
ARCHITECTURE gate_level OF bit_comparator IS 
FOR ALL : n1 USE ENTITY WORK.inv (single_delay);


 
FOR ALL : n2 USE ENTITY WORK.nand2 (single_delay);

 
FOR ALL : n3 USE ENTITY WORK.nand3 (single_delay);

 
-- Intermediate signals 
SIGNAL im1,im2, im3, im4, im5, im6, im7, im8, im9, im10 : BIT; 
BEGIN 
-- a_gt_b output 
g0 : n1 PORT MAP (a, im1);              g1 : n1 PORT MAP (b, im2);
 
g2 : n2 PORT MAP (a, im2, im3);    g3 : n2 PORT MAP (a, gt, im4); 
g4 : n2 PORT MAP (im2, gt, im5);






 
g5 : n3 PORT MAP (im3, im4, im5, a_gt_b);




 
-- a_eq_b output 
g6 : n3 PORT MAP (im1, im2, eq, im6); 






g7 : n3 PORT MAP (a, b, eq, im7);






 
g8 : n2 PORT MAP (im6, im7, a_eq_b);





 
-- a_lt_b output 
g9 : n2 PORT MAP (im1, b, im8); g10 : n2 PORT MAP (im1, lt, im9);
 
g11 : n2 PORT MAP (b, lt, im10);






 
g12 : n3 PORT MAP (im8, im9, im10, a_lt_b);




 
END gate_level; 

Using Component Package


USE WORK.simple_gates.ALL  ;
-- is equaivalent to: 

USE 
WORK.simple_gates.n1  , 
WORK.simple_gates.n2  , 
WORK.simple_gates.n3  ; 
-- n1, n2 and n3 component declarations are visible 



· Apply USE to make a package visible 

· WORK is the working library of the component package 

· Both forms shown above are equivalent 

· If a Package contains No Subprograms, Package Body is not Required in VHDL-87  but is  Required in VHDL-93.
Utility Package

PACKAGE basic_utilities IS 

TYPE integers IS ARRAY (0 TO 12) OF INTEGER; 
FUNCTION fgl (w, x, gl : BIT) RETURN BIT; 
FUNCTION feq (w, x, eq : BIT) RETURN BIT; 
PROCEDURE bin2int (bin : IN BIT_VECTOR; int : OUT Integer); PROCEDURE int2bin (int : IN INTEGER; bin : OUT Bit_Vector); 
PROCEDURE apply_data (Signal target : out Bit_Vector (3 DownTo 0); Constant values : in integers; Constant period : in TIME); 
END basic_utilities; 

PACKAGE BODY basic_utilities IS 

FUNCTION fgl (w, x, gl : BIT) RETURN BIT IS 

BEGIN 

RETURN (w AND gl) OR (NOT x AND gl) OR (w AND NOT x); 

END fgl; 

FUNCTION feq (w, x, eq : BIT) RETURN BIT IS 
BEGIN 

RETURN (w AND x AND eq) OR (NOT w AND NOT x AND eq); 
END feq; 

Function bin2int (bin : IN Bit_Vector; int : OUT Integer) Return IS 
Variable result: INTEGER; 

Variable SUM: INTEGER := 1;

BEGIN
result := 0;

For i in Bin’LOW To Bin’High LOOP 

IF Bin(i) = '1' then  Result:= Result + SUM ; end if;

SUM :=2*SUM;

END LOOP;

Return result;

END bin2int;

PROCEDURE int2bin (int : IN Integer; bin : out BIT_VECTOR) IS 
VARIABLE tmp : INTEGER;

BEGIN 

tmp := int; 

FOR i IN bin'Low TO bin'High LOOP 
IF (tmp MOD 2 = 1) THEN bin (i) := '1'; ELSE bin (i) := '0'; END IF;

tmp := tmp / 2; 
END LOOP; 
END int2bin; 

PROCEDURE apply_data (Signal target : out Bit_Vector (3 DownTo 0); Constant values : in integers; Constant period : in Time) is
Variable buf : Bit_Vector (3 DOWNTO 0);

BEGIN 

FOR i IN values’Low TO values’high LOOP
int2bin (values(i), buf);

target <= TRANSPORT buf AFTER i * period;

END LOOP;

END apply_data;

END basic_utilities; 

Design_Organization.packaging.utilities

USE WORK.basic_utilities.ALL;








ARCHITECTURE functional OF bit_comparator IS 

BEGIN

a_gt_b <= fgl (a, b, gt) AFTER 12 NS;

a_eq_b <= feq (a, b, eq) AFTER 12 NS; 

a_lt_b <= fgl (b, a, lt) AFTER 12 NS; 

END functional; 

USE WORK.basic_utilities.ALL; 

ARCHITECTURE procedural OF nibble_comparator_test_bench is
Component comp4 PORT (a, b : in bit_vector (3 DOWNTO 0); gt, eq, lt : in BIT; a_gt_b, a_eq_b, a_lt_b : out BIT); 

END Component; 

For a1 : comp4 use BEGIN work.nibble_comparator(structural);

SIGNAL a, b : BIT_VECTOR (3 DOWNTO 0); 

SIGNAL eql, lss, gtr : BIT; 

SIGNAL vdd : BIT := '1';                  SIGNAL gnd : BIT := '0';

BEGIN 

a1: comp4 PORT MAP (a, b, gnd, vdd, gnd, gtr, eql, lss); 

apply_data 

    (a, 0&15&15&14&14&14&14&10&00&15&00&00&15, 500 ns);

apply_data 

     (b, 0&14&14&15&15&12&12&12&15&15&15&00&00, 500 ns);

END procedural; 

· USE statement makes basic_utilities package visible 

Design_Parametrization

ENTITY   inv_t   IS
GENERIC (tplh : TIME := 3 NS; tphl : TIME := 5 NS);

PORT (i1 : in BIT; o1 : out BIT);

END inv_t;

--

ARCHITECTURE average_delay OF inv_t IS

BEGIN
o1 <= NOT i1 AFTER (tplh + tphl) / 2;

END average_delay; 

ENTITY nand2_t IS

GENERIC (tplh : TIME := 4 NS; tphl : TIME := 6 NS);

PORT (i1, i2 : IN BIT; o1 : OUT BIT); 

END nand2_t;

-- 

ARCHITECTURE average_delay OF nand2_t IS 

BEGIN 

o1 <= i1 NAND i2 AFTER (tplh + tphl) / 2; 

END average_delay; 



· GENERICs can pass design parameters 

· New versions of gate descriptions contain timing 

· GENERICs can include default values 

ENTITY nand3_t IS 

GENERIC (tplh : TIME := 5 NS; tphl : TIME := 7 NS); 

PORT (i1, i2, i3 : IN BIT; o1 : OUT BIT); 

END nand3_t;

--

ARCHITECTURE average_delay OF nand3_t IS 

BEGIN
o1 <= NOT ( i1 AND i2 AND i3 ) AFTER (tplh + tphl) / 2;

END average_delay; 

Syntax

· GENERIC clause has syntax similar to PORT clause 

· Interface CONSTANT declaration 

Values Passed to Generic Parameters  

Several alternatives for passing values to GENERICs 

1. Using defaults 

2. Assigning fixed values 

3. Passing values from higher level components 

4. Configuring designs 


1. Using Default_Values 

ARCHITECTURE default_delay OF bit_comparator IS

Component n1 PORT (i1: IN BIT; o1: OUT BIT);

END Component;
Component n2 PORT (i1, i2: IN BIT; o1: OUT BIT);

END Component;
Component n3 PORT (i1, i2, i3: IN BIT; o1: OUT BIT);

END Component;
FOR ALL : n1 USE ENTITY WORK.inv_t (average_delay);

FOR ALL : n2 USE ENTITY WORK.nand2_t (average_delay);

FOR ALL : n3 USE ENTITY WORK.nand3_t (average_delay);

-- Intermediate signals

SIGNAL im1,im2, im3, im4, im5, im6, im7, im8, im9, im10 : BIT;

BEGIN
-- a_gt_b output

g0 : n1 PORT MAP (a, im1); 

g1 : n1 PORT MAP (b, im2);

g2 : n2 PORT MAP (a, im2, im3); 

g3 : n2 PORT MAP (a, gt, im4);

g4 : n2 PORT MAP (im2, gt, im5);

g5 : n3 PORT MAP (im3, im4, im5, a_gt_b);

-- a_eq_b output 

g6 : n3 PORT MAP (im1, im2, eq, im6);

g7 : n3 PORT MAP (a, b, eq, im7);

g8 : n2 PORT MAP (im6, im7, a_eq_b);

-- a_lt_b output

g9 : n2 PORT MAP (im1, b, im8); g10 : n2 PORT MAP (im1, lt, im9);

g11 : n2 PORT MAP (b, lt, im10);

g12 : n3 PORT MAP (im8, im9, im10, a_lt_b);

END default_delay; 

· Component declarations do not contain GENERICs 

· Component instantiation are as before 

· If default values exits, they are used 

2. Assigning Fixed Values to Generic Parameters

ARCHITECTURE fixed_delay OF bit_comparator IS
Component n1
Generic (tplh, tphl : Time); Port (i1: in Bit; o1: out Bit);

END Component;
Component n2

Generic (tplh, tphl : Time); Port (i1, i2: in Bit; o1: out Bit);

END Component;
Component n3
Generic (tplh, tphl : Time); Port (i1, i2, i3: in Bit; o1: out Bit);

END Component;
FOR ALL : n1 USE ENTITY WORK.inv_t (average_delay);

FOR ALL : n2 USE ENTITY WORK.nand2_t (average_delay);

FOR ALL : n3 USE ENTITY WORK.nand3_t (average_delay);

-- Intermediate signals

SIGNAL im1,im2, im3, im4, im5, im6, im7, im8, im9, im10 : BIT; 

BEGIN
-- a_gt_b output

g0 : n1 Generic Map (2 NS, 4 NS) Port Map (a, im1);

g1 : n1 Generic Map (2 NS, 4 NS) Port Map (b, im2);

g2 : n2 Generic Map (3 NS, 5 NS) Port Map (a, im2, im3);

g3 : n2 Generic Map (3 NS, 5 NS) Port Map P (a, gt, im4); 

g4 : n2 Generic Map (3 NS, 5 NS) Port Map (im2, gt, im5); 

g5 : n3 Generic Map (4 NS, 6 NS) Port Map (im3, im4, im5, a_gt_b);
-- a_eq_b output
g6 : n3 Generic Map (4 NS, 6 NS) Port Map (im1, im2, eq, im6);

-- a_lt_b output

g9 : n2 Generic Map (3 NS, 5 NS) Port Map (im1, b, im8);

END fixed_delay; 

· Component declarations contain GENERICs 

· Component instantiation contain GENERIC Values 

· GENERIC Values overwrite default values 



3. Passing_Values  From Higher Level Specs




ENTITY bit_comparator_t IS

GENERIC (tplh1, tplh2, tplh3, tphl1, tphl2, tphl3 : TIME);

PORT (a, b, gt, eq, lt : IN BIT; a_gt_b, a_eq_b, 

              a_lt_b : OUT BIT); 

END bit_comparator_t; 



· To pass values, upper level units must contain GENERICs 

· A timed bit_comparator is developed 

ARCHITECTURE passed_delay OF bit_comparator_t IS 

Component n1

Generic (tplh, tphl : Time); Port (i1: in Bit; o1: out Bit);

END Component;
Component n2

Generic (tplh, tphl : Time); Port (i1, i2: in Bit; o1: out Bit); 

END Component;
Component n3
Generic (tplh, tphl : Time); Port (i1, i2, i3: in Bit; o1: out Bit);

END Component; 

FOR ALL : n1 USE ENTITY WORK.inv_t (average_delay);

FOR ALL : n2 USE ENTITY WORK.nand2_t (average_delay);

FOR ALL : n3 USE ENTITY WORK.nand3_t (average_delay);

-- Intermediate signals

SIGNAL im1,im2, im3, im4, im5, im6, im7, im8, im9, im10 : Bit;

BEGIN
-- a_gt_b output

g0 : n1 Generic Map (tplh1, tphl1) Port Map (a, im1);

g1 : n1 Generic Map (tplh1, tphl1) Port Map (b, im2);

g2 : n2 Generic Map (tplh2, tphl2) Port Map (a, im2, im3);

g3 : n2 Generic Map (tplh2, tphl2) Port Map (a, gt, im4);

g4 : n2 Generic Map (tplh2, tphl2) Port Map (im2, gt, im5);

g5 : n3 Generic Map (tplh3, tphl3) Port Map (im3, im4, im5, a_gt_b);

-- a_eq_b output

g6 : n3 Generic Map (tplh3, tphl3) Port Map (im1, im2, eq, im6);

-- a_lt_b output 

g9 : n2 Generic Map (tplh2, tphl2) Port Map (im1, b, im8); 

END passed_delay; 



· Component declarations include GENERICs 

· Component instantiations include passed values 

· GENERIC maps are required 



4. Passing_Values Through Component Defaults

ARCHITECTURE iterative OF nibble_comparator IS
Component comp1

Generic (tplh1 : Time := 2 ns; tplh2 : Time:= 3 ns; tplh3 : Time:= 4 ns; tphl1 : Time:= 4 ns; tphl2 : Time:= 5 ns; tphl3 : Time:= 6 ns); 

Port (a, b, gt, eq, lt : in Bit; a_gt_b, a_eq_b, a_lt_b : Out Bit); 

END Component;
FOR ALL : comp1 USE ENTITY WORK.bit_comparator_t (passed_delay);

SIGNAL im : BIT_VECTOR ( 0 TO 8);

BEGIN
c0: comp1 Port Map (a(0), b(0), gt, eq, lt, im(0), im(1), im(2)); 

c1to2: FOR i IN 1 TO 2 GENERATE 

c: comp1 PORT MAP (a(i), b(i), im(i*3-3), im(i*3-2), im(i*3-1), im(i*3+0), im(i*3+1), im(i*3+2) );

END GENERATE;

c3: comp1 PORT MAP (a(3), b(3), im(6), im(7), im(8), a_gt_b, a_eq_b, a_lt_b); 

END iterative; 



· Using bit_comparator_t in a larger design 

· Specification of GENERIC parameters is Required 

· Exclusion of GENERIC Map leaves all parameters OPEN 
· Association with OPEN causes default values to be used 



Instances with OPEN Parameter Association

ARCHITECTURE iterative OF nibble_comparator IS 

BEGIN
c0: comp1

GENERIC MAP (Open, Open, 8 NS, Open, Open, 10 NS)

PORT MAP (a(0), b(0), gt, eq, lt, im(0), im(1), im(2)); 
………………….

END iterative; 

ARCHITECTURE iterative OF nibble_comparator IS

………………….

BEGIN
c0: comp1

GENERIC MAP (tplh3 => 8 NS, tphl3 => 10 NS)

PORT MAP (a(0), b(0), gt, eq, lt, im(0), im(1), im(2));

……………………

END iterative; 



· A GENERIC Map May specify only some of the parameters 

· Using OPEN causes use of default Component Values 

· Alternatively, association by name can be used 

· Same applies to PORT MAP 
Configuration Declarations

USE WORK.basic_utilities.ALL; -- Grants Entities Visibility 

ARCHITECTURE customizable OF nibble_comparator_test_bench IS 

--

Component comp4 PORT (a, b : IN bit_vector (3 DOWNTO 0); gt, eq, lt : IN BIT; a_gt_b, a_eq_b, a_lt_b : OUT BIT);

END Component; 

--

SIGNAL a, b : BIT_VECTOR (3 DOWNTO 0); 

SIGNAL eql, lss, gtr : BIT;

SIGNAL vdd : BIT := '1'; 

SIGNAL gnd : BIT := '0';

--

BEGIN 

a1: comp4 PORT MAP (a, b, gnd, vdd, gnd, gtr, eql, lss);

apply_data (a, 00&15&15&14&14&14&14&10&00&15&00&00&15, 500 NS);

apply_data (b, 00&14&14&15&15&12&12&12&15&15&15&00&00, 500 NS); 

END customizable; 



A general purpose testbench 

(The comp4 component is unbound )

Configure it to test any of the previous nibble_comparators 

Customizable Testbench Configuration Declaration

Configuration functional OF nibble_comparator_test_bench IS
FOR customizable 

FOR a1 : comp4

USE ENTITY WORK.nibble_comparator (structural);

END FOR;

END FOR;

END functional; 

CONFIGURATION average_delay OF nibble_comparator_test_bench IS
FOR customizable

FOR a1 : comp4

USE ENTITY WORK.nibble_comparator (iterative);

END FOR;

END FOR;

END average_delay; 

----------------------------------------------------------------------------

· Configuration of testbench requires binding of a1:comp4 
· Bind a1:comp4 to a nibble_comparator architechture 

· Testbench configured for 

1) Structural ARCHITECTURE of nibble_comparator 
2) Iterative ARCHITECTURE of nibble_comparator 

SYNTAX

· Syntax details of configuration declaration 

· Block configuration gives visibility 

· Component configuration configures a component 

Configuration Nesting

ARCHITECTURE flexible OF nibble_comparator IS 

Component comp1

Port (a, b, gt, eq, lt : in Bit; a_gt_b, a_eq_b, a_lt_b : out Bit);

END Component; 

SIGNAL im : BIT_VECTOR ( 0 TO 8 );

--

BEGIN 

c0: comp1 Port Map (a(0), b(0), gt, eq, lt, im(0), im(1), im(2));

c1to2: FOR i IN 1 TO 2 GENERATE
c: comp1 Port Map (a(i), b(i), im(i*3-3), im(i*3-2), im(i*3-1), im(i*3+0), im(i*3+1), im(i*3+2) );

END GENERATE; 

c3: comp1 Port Map (a(3), b(3), im(6), im(7), im(8), a_gt_b, a_eq_b, a_lt_b);

END flexible; 



Will demonstrate nesting by two levels of flexibility 

For Flexible nibble_comparator 
· No particular bit_comparator is specified 

· Using customizable testbench creates flexibility for: 

1) Specifying nibble_comparator for testbench 

2) Specifying bit_comparator for nibble_comparator

Configuration Nesting

USE WORK.ALL;  -- Grant Entities Visibility
________________________________________________________

Configuration default_bit_level Of nibble_comparator_test_bench IS
FOR customizable
FOR a1 : comp4

USE ENTITY WORK.nibble_comparator(flexible); 

FOR flexible
FOR c0, c3: comp1

 
USE ENTITY WORK.bit_comparator (default_delay);

END FOR;

FOR c1to2  -- Grants Generate Block Visibility
FOR c: comp1

USE ENTITY WORK.bit_comparator (default_delay);

END FOR;

END FOR; 

END FOR;

END FOR;

END FOR;

END default_bit_level; 

___________________________________________________

· Configuring customizable testbench for testing default_delay bit_comparator  instantiated through  flexible nibble_comparator 

· Visibility is obtained before component configuration is done 

Configurations and .Generics

USE WORK.ALL;  -- Grant Entities Visibility
Configuration passed_bit_level Of nibble_comparator_test_bench IS
FOR customizable
FOR a1 : comp4

USE ENTITY WORK.nibble_comparator(flexible); 

FOR flexible
FOR c0, c3: comp1

 
USE ENTITY work.bit_comparator_t (Passed_delay)

Generic Map (tplh1 => 2 NS, tplh2 => 3 NS, tplh3 => 4 NS, tphl1 => 4 NS, tphl2 => 5 NS, tphl3 => 6 NS); 
END FOR;

FOR c1to2  -- Grants Generate Block Visibility
FOR c: comp1


USE ENTITY work.bit_comparator_t (Passed_delay)

Generic Map (tplh1 => 2 NS, tplh2 => 3 NS, tplh3 => 4 NS, tphl1 => 4 NS, tphl2 => 5 NS, tphl3 => 6 NS); 
END FOR;

END FOR; 

END FOR;

END FOR;

END FOR;

END passed _bit_level; 



bit_comparator_t uses GENERIC parameters, Configure for: 

· Component binding, and 

· GENERIC Map specification 

Same can be done with PORT MAP 
n_bit_Register Example

· NAND gates form sr_latch 
· Add an inverter to form d_latch 
· n d_latch structures form an n-bit register 

___________________________________________________

ENTITY sr_latch IS PORT (s, r, c : IN BIT; q : OUT BIT); 

END sr_latch;

ARCHITECTURE gate_level OF sr_latch IS

Component n2 Port (i1, i2: IN Bit; o1: OUT Bit); End Component; 

SIGNAL im1, im2, im3, im4 : BIT; 

BEGIN
g1 : n2 Port Map (s, c, im1); g2 : n2 Port Map (r, c, im2);

g3 : n2 Port Map (im1, im4, im3); g4 : n2 Port Map (im3, im2, im4);

q <= im3;

END gate_level; 

ENTITY d_latch IS PORT (d, c : IN BIT; q : OUT BIT); END d_latch; 

ARCHITECTURE sr_based OF d_latch IS

Component sr Port (s, r, c : in Bit; q : out Bit); End Component; 

Component n1 Port (i1: in Bit; o1: out Bit); End Component;
SIGNAL dbar : BIT;

BEGIN 

c1 : sr Port Map (d, dbar, c, q); 

c2 : n1 Port Map (d, dbar);

END sr_based; 

___________________________________________________

ENTITY d_register IS

PORT (d : in Bit_Vector; c : in Bit; q : out Bit _Vector );

END d_register;

--

ARCHITECTURE latch_based OF d_register IS

Component dl PORT (d, c : in Bit; q : out Bit); End Component; 
BEGIN
dn : FOR i IN d'RANGE GENERATE 

di : dl PORT MAP (d(i), c, q(i));

END GENERATE; 

END latch_based; 



· gate_level sr_latch uses unbound gate components 

· sr_based d_latch uses unbound sr_latch and gate 

· latch_based d_register uses unbound d_latch components 

· Different delay values remedy the oscillation problem 

Configuration of the n_bit_register

USE WORK.ALL; 

CONFIGURATION average_gate_delay OF d_register IS 

FOR latch_based –Architecture name 

FOR dn – Generate Visibility 

FOR di : dl --  

USE ENTITY WORK.d_latch(sr_based); 

FOR sr_based --5 
FOR c1 : sr --6 
  USE ENTITY WORK.sr_latch(gate_level);

FOR gate_level –7
FOR g2, g4 : n2 USE ENTITY WORK.nand2_t (average_delay) 

GENERIC MAP (5 NS, 6 NS);

END FOR; 

FOR g1, g3 : n2 USE ENTITY WORK.nand2_t (average_delay)

GENERIC MAP (2 NS, 4 NS);

END FOR;

END FOR;

END FOR;

FOR c2 : n1 

USE ENTITY WORK.inv_t(average_delay)

GENERIC MAP (3 NS, 5 NS);

END FOR;

END FOR;

END FOR;

END FOR;

END FOR;

END average_gate_delay; 



· To avoid timing problem: g2,g4 delays are at 5NS, 6NS while g1,g3 delays are at 2NS, 4NS 

Alternative Configuration of the n_bit_register
USE WORK.ALL; 

CONFIGURATION single_gate_delay OF d_register IS

FOR latch_based –Architecture name 

FOR dn – Generate Visibility 

FOR di : dl --  

USE ENTITY WORK.d_latch(sr_based); 

FOR sr_based --5 
FOR c1 : sr --6 
  USE ENTITY WORK.sr_latch(gate_level);

FOR gate_level –7
FOR g2, g4 : n2 USE Entity Work.nand3 (single_delay) 

   Port Map (i1, i1, i2, o1); 

End For;

For g1, g3 : n2 USE Entity Work.nand2 (single_delay);End For; 

END FOR;

END FOR;

FOR c2 : n1 USE ENTITY WORK.inv (single_delay);

END FOR;

END FOR;

END FOR;

END FOR;

END FOR; 

END single_gate_delay; 



· For g2 and g4 use NAND3 in place of NAND2 

· This also eliminates the timing problem 

· PORT MAP overwrites default in component declaration 

Test Bench of 8_bit_register

ARCHITECTURE single OF d_register_test_bench IS 

Component reg Port (d : in Bit_Vector (7 Downto 0); C : in Bit; q : Out Bit_Vector (7 Downto 0) ); End Component; 
--

For r8 : reg Use Configuration Work.single_gate_delay;

--

SIGNAL data, outdata : BIT_VECTOR (7 Downto 0);

SIGNAL clk : BIT; 

BEGIN
r8: reg   Port Map (data, clk, outdata);

data <= X"00", X"AA" After 0500 NS, X"55" After 1500 NS; 

clk <= 
'0', '1' After 0200 NS, '0' After 0300 NS,

'1' AFTER 0700 NS, '0' AFTER 0800 NS, 

'1' AFTER 1700 NS, '0' AFTER 1800 NS; 

END single; 



· Instantiating n-bit register in a testbench 

· Associating d with an 8 bit vector makes an 8-bit register 

Selective Configuration  Of Generate-Statement Instances

Example Parity checker

ENTITY xor2_t IS

GENERIC (tplh : TIME := 9 NS; tphl : TIME := 7 NS);

PORT (i1, i2 : IN BIT; o1 : OUT BIT);

END xor2_t;

--

ARCHITECTURE average_delay OF xor2_t IS

BEGIN 
o1 <= i1 XOR i2 AFTER (tplh + tphl) / 2;

END average_delay;

---- 

ENTITY inv_t IS

GENERIC (tplh : TIME := 5 NS; tphl : TIME := 3 NS);

PORT (i1 : IN BIT; o1 : OUT BIT);

END inv_t; 

--

ARCHITECTURE average_delay OF inv_t IS

BEGIN
o1 <= NOT i1 AFTER (tplh + tphl) / 2;

END average_delay; 



· Develop an eight bit parity checker 

· Use an XOR gate and an inverter 

ENTITY parity IS

Port (A : In Bit_Vector (7 Downto 0); Odd, Even : Out Bit);

END parity;

--

ARCHITECTURE iterative OF parity IS

Component X2 Port (I1, I2: In Bit; O1: Out Bit); End Component; 

Component N1 Port (I1: In Bit; O1: Out Bit); End Component; 

SIGNAL im : BIT_VECTOR ( 0 TO 6 );

BEGIN 

first: x2 PORT MAP (a(0), a(1), im(0));

middle: FOR i IN 1 TO 6 GENERATE 

m: x2 PORT MAP (im(i-1), a(i+1), im(i)); 

      END GENERATE; 

last: odd <= im(6); 

inv: n1 PORT MAP (im(6), even); 

END iterative; 

CONFIGURATION parity_binding OF parity IS

FOR iterative

FOR first : x2

USE ENTITY WORK.xor2_t (average_delay) 

GENERIC MAP (5 NS, 5 NS);

END FOR;

FOR middle(1 TO 5)

FOR m : x2

USE ENTITY WORK.xor2_t (average_delay)

GENERIC MAP (5 NS, 5 NS);

END FOR;

END FOR;

FOR middle ( 6)

FOR m : x2

USE ENTITY WORK.xor2_t (average_delay)

GENERIC MAP (6 NS, 7 NS);

END FOR;

END FOR;

FOR inv : n1

USE ENTITY WORK.inv_t (average_delay) GENERIC MAP (5 NS, 5 NS);

END FOR;

END FOR;

END parity_binding; 



· Due to fanout, last gate has a higher delay 

· Element 6 of Generate statement specifies 6 NS, 7 NS delays 

· Other generated elements use 5 NS, 5 NS 
· Generate Index Value May be Used to Bind different instances of the Generate Statement to Different Entities/Architectures

· Can use OTHERS for indexing all other instantiations 

Design Libraries

· A Design Library is A Set of Pre-Compiled Pre-Analyzed Design Units.
· Design Units:

· Entities, 

· Architecture Bodies,

· Package Declarations, 

· Package Bodies, and 

· Configurations

· Types of Libraries:

Two Types of Libraries

1. Working Library  (WORK) {A Predefined library into which a Design Unit is Placed after Compilation.}, 

2. Resource Libraries  {Contain design units that can be referenced within the design unit being compiled}. 

· Only one library can be the Working Library

· Any number of Resource Libraries May be Used by a Design Entity

· There is a Number of Predefined Resource Libraries

· The Library Clause is Used To Make A Given Library Visible
· The Use-Clause Causes Package  Declarations Within a Library to be Visible
· Library Management Tasks, e.g. Creation or Deletion, are not Part of the VHDL Language Standard ( Tool Dependent
· Predifined Libraries:

· The STD Library Contains 2 Packages Standard and Textio (See Appendix F in the Textbook).

· The Standard Package Contains All the Predefined Data Types, e.g. BIT, BIT_Vector, Character, Integer, Boolean, etc. 

· The Textio Package Contains some Utilities for reading and writing of Data into Files

· By Default, Every Design Unit is Assumed to Contain the Following Declarations:

LIBRARY 
STD , work ; 
USE 


STD.Standard.All ;
· Another Standard Library is the  IEEE Library which Contains the Standard Package Std_Logic_1164 which is becoming a De Facto Standard for all Synthesis Tools (See Appendix G in the Textbook).

· The Std_Logic_1164 Package Defines a new Logic System consisting of 9-Valued Data Type and Related Utility Functions and Procedures For This 9-Vlued System.

· Subtypes of this 9-Valued Are Also Defined Together with Overloaded Functions and Operations on these Subtypes

          LIBRARY   IEEE ; 
                 USE 

   IEEE.STD_Logic_1164.All ;
STD_LOGIC_1164 9-Valued Logic System

TYPE std_ulogic IS 
(
'U', -- Uninitialized {Important For Sequential Systems}

'X', -- Forcing Unknown {Contention}

'0', -- Forcing 0

'1', -- Forcing 1

'Z', -- High Impedance

'W', -- Weak Unknown

'L', -- Weak 0  {,e.g. Logic 0 Held Dynamically on a Capacitor}

'H', -- Weak 1  {,e.g. Logic1 Held Dynamically on a Capacitor}

'-' -- Don't care{Used To Optimize Synthesis}

); 
TYPE std_ulogic_vector IS Array ( Natural Range <> ) OF std_ulogic ; 
Resolving Multiple Values on the Same Signal Driver:

1. Strong Dominates Weak

2. Weak Dominates Z

3. Conflicts of Equal Strength is Unknown at That Strength

TYPE stdlogic_table IS ARRAY(std_ulogic, std_ulogic) OF std_ulogic ;
    -- resolution function

    ----------------------------------    

CONSTANT resolution_table : stdlogic_table
    := (
  ---------------------------------

-- |  U    X    0    1    Z    W    L    H    -        |   |  

    ---------------------------------

   ( 'U', 'U', 'U', 'U', 'U', 'U', 'U', 'U', 'U' ), -- | U |
   ( 'U', 'X', 'X', 'X', 'X', 'X', 'X', 'X', 'X' ), -- | X |
   ( 'U', 'X', '0', 'X', '0', '0', '0', '0', 'X' ), -- | 0 |
   ( 'U', 'X', 'X', '1', '1', '1', '1', '1', 'X' ), -- | 1 |
   ( 'U', 'X', '0', '1', 'Z', 'W', 'L', 'H', 'X' ), -- | Z |
   ( 'U', 'X', '0', '1', 'W', 'W', 'W', 'W', 'X' ), -- | W |
   ( 'U', 'X', '0', '1', 'L', 'W', 'L', 'W', 'X' ), -- | L |
   ( 'U', 'X', '0', '1', 'H', 'W', 'W', 'H', 'X' ), -- | H |
   ( 'U', 'X', 'X', 'X', 'X', 'X', 'X', 'X', 'X' )  -- | - |
        );

FUNCTION resolved ( s : std_ulogic_vector ) RETURN std_ulogic  IS
 VARIABLE result : std_ulogic := 'Z';  -- weakest state default
    BEGIN

        -- the test for a single driver is essential otherwise the

        -- loop would return 'X' for a single driver of '-' and that

        -- would conflict with the value of a single driver unresolved

        -- signal.

        IF    (s'LENGTH = 1) THEN    RETURN s(s'LOW); ELSE

            FOR i IN s'RANGE LOOP
                result := resolution_table (result, s(i));

            END LOOP;

        END IF;

        RETURN result;

    END resolved;

SUBTYPE std_logic IS  resolved  std_ulogic ; 

TYPE std_logic_vector Is Array ( Natural Range <>) OF std_logic ; 

· std_ulogic  is an Unresolved  9-Valued System 

· The resolved function resolves std_ulogic Values 

· The std_logic  and std_logic_vector are de-facto industry standard 

Subsets of the std_logic Data Type

SUBTYPE X01 IS resolved std_ulogic RANGE 'X' TO '1'; 

   -- ('X','0','1') 
SUBTYPE X01Z IS resolved std_ulogic RANGE 'X' TO 'Z'; 

--('X','0','1','Z') 
SUBTYPE UX01 IS resolved std_ulogic RANGE 'U' TO '1'; 

-- (‘U’,'X','0','1') 
SUBTYPE UX01Z IS resolved std_ulogic RANGE 'U' TO 'Z'; 

-- (‘U’,'X','0','1', ‘Z’) 
· Subtypes provide mapping to common value systems without Strength  X01, X01Z, UX01, UX01Z 

· The std_logic is a Superset of most common value systems 

-- Overloaded Logical Operators

    -------------------------------------------------------------------    

FUNCTION "and"  ( l : std_ulogic; r : std_ulogic ) RETURN UX01;

FUNCTION "nand" ( l : std_ulogic; r : std_ulogic ) RETURN UX01;

FUNCTION "or"   ( l : std_ulogic; r : std_ulogic ) RETURN UX01;

FUNCTION "nor"  ( l : std_ulogic; r : std_ulogic ) RETURN UX01;

FUNCTION "xor"  ( l : std_ulogic; r : std_ulogic ) RETURN UX01;

FUNCTION "xnor" ( l : std_ulogic; r : std_ulogic ) return ux01;

FUNCTION "not"  ( l : std_ulogic                 ) RETURN UX01;

    -------------------------------------------------------------------    

-- Vectorized Overloaded Logical Operators

    -------------------------------------------------------------------    

FUNCTION "and"  ( l, r : std_logic_vector  ) RETURN std_logic_vector;

FUNCTION "and"  ( l, r : std_ulogic_vector ) RETURN std_ulogic_vector;

FUNCTION "nand" ( l, r : std_logic_vector  ) RETURN std_logic_vector;

FUNCTION "nand" ( l, r : std_ulogic_vector ) RETURN std_ulogic_vector;

FUNCTION "or"   ( l, r : std_logic_vector  ) RETURN std_logic_vector;

FUNCTION "or"   ( l, r : std_ulogic_vector ) RETURN std_ulogic_vector;

FUNCTION "nor"  ( l, r : std_logic_vector  ) RETURN std_logic_vector;

FUNCTION "nor"  ( l, r : std_ulogic_vector ) RETURN std_ulogic_vector;

FUNCTION "xor"  ( l, r : std_logic_vector  ) RETURN std_logic_vector;

FUNCTION "xor"  ( l, r : std_ulogic_vector ) RETURN std_ulogic_vector;

Function "xnor" ( l, r : std_logic_vector  ) return std_logic_vector;

Function "xnor" ( l, r : std_ulogic_vector ) return std_ulogic_vector;

FUNCTION "not"  ( l : std_logic_vector  ) RETURN std_logic_vector;

FUNCTION "not"  ( l : std_ulogic_vector ) RETURN std_ulogic_vector;

    -------------------------------------------------------------------

-- Conversion Functions

    -------------------------------------------------------------------FUNCTION To_bit       ( s : std_ulogic;        xmap : BIT := '0') RETURN BIT;

FUNCTION To_bitvector ( s : std_logic_vector ; xmap : BIT := '0') RETURN BIT_VECTOR;

FUNCTION To_bitvector ( s : std_ulogic_vector; xmap : BIT := '0') RETURN BIT_VECTOR;

FUNCTION To_StdULogic   ( b : BIT   ) RETURN std_ulogic;

FUNCTION To_StdLogicVector  ( b : BIT_VECTOR        ) RETURN std_logic_vector;

FUNCTION To_StdLogicVector  ( s : std_ulogic_vector ) RETURN std_logic_vector;

FUNCTION To_StdULogicVector ( b : BIT_VECTOR        ) RETURN std_ulogic_vector;

FUNCTION To_StdULogicVector ( s : std_logic_vector  ) RETURN std_ulogic_vector;

    -------------------------------------------------------------------    

-- strength strippers and type convertors

    -------------------------------------------------------------------    

FUNCTION To_X01  ( s : std_logic_vector  ) RETURN  std_logic_vector;

FUNCTION To_X01  ( s : std_ulogic_vector ) RETURN  std_ulogic_vector;

FUNCTION To_X01  ( s : std_ulogic        ) RETURN  X01;

FUNCTION To_X01  ( b : BIT_VECTOR        ) RETURN  std_logic_vector;

FUNCTION To_X01  ( b : BIT_VECTOR        ) RETURN  std_ulogic_vector;

FUNCTION To_X01  ( b : BIT               ) RETURN  X01;       

FUNCTION To_X01Z ( s : std_logic_vector  ) RETURN  std_logic_vector;

FUNCTION To_X01Z ( s : std_ulogic_vector ) RETURN  std_ulogic_vector;

FUNCTION To_X01Z ( s : std_ulogic        ) RETURN  X01Z;

FUNCTION To_X01Z ( b : BIT_VECTOR        ) RETURN  std_logic_vector;

FUNCTION To_X01Z ( b : BIT_VECTOR        ) RETURN  std_ulogic_vector;

FUNCTION To_X01Z ( b : BIT               ) RETURN  X01Z;      

FUNCTION To_UX01  ( s : std_logic_vector  ) RETURN  std_logic_vector;

FUNCTION To_UX01  ( s : std_ulogic_vector ) RETURN  std_ulogic_vector;

FUNCTION To_UX01  ( s : std_ulogic        ) RETURN  UX01;

FUNCTION To_UX01  ( b : BIT_VECTOR        ) RETURN  std_logic_vector;

FUNCTION To_UX01  ( b : BIT_VECTOR        ) RETURN  std_ulogic_vector;

FUNCTION To_UX01  ( b : BIT               ) RETURN  UX01;       

    -------------------------------------------------------------------    

-- Edge Detection

    -------------------------------------------------------------------    

FUNCTION rising_edge  (SIGNAL s : std_ulogic) RETURN BOOLEAN;

FUNCTION falling_edge (SIGNAL s : std_ulogic) RETURN BOOLEAN;

    -------------------------------------------------------------------    

-- object contains an unknown

    -------------------------------------------------------------------    

FUNCTION Is_X ( s : std_ulogic_vector ) RETURN  BOOLEAN;

FUNCTION Is_X ( s : std_logic_vector  ) RETURN  BOOLEAN;

FUNCTION Is_X ( s : std_ulogic        ) RETURN  BOOLEAN;



· Overloading logical operators 

· Vectorized operators are overloaded 

· Vectorized operators are for resolved and unresolved types 

· Conversion functions for common logic value systems 
· Conversion to systems without strength 

· Vectorized conversion functions are included 

LIBRARY "ls7400" User: ... Date: ... 

simple_gates PACKAGE DECLARATION ... 

inv ENTITY ... 

inv (single_delay) ARCHITECTURE ... 

nand2 ENTITY ... 

nand2 (single_delay) ARCHITECTURE ... 

nand3 ENTITY ... 

nand3 (single_delay) ARCHITECTURE ... 



A library can hold 

PACKAGE DECLARATIONS 

PACKAGE BODY 

ENTITY 

ARCHITECTURE 

Library managment is platform dependent 



Design_Organization.libraries 

WORK is the default working library 

WORK is always visible 

VHDL includes STD library 

STD contains STANDARD and TEXTIO package 

STD library is always visible 

Other user libraries can be created 

Library utilization: 

To specify use of a library: 

LIBRARY my_lib 
To specify use of all library contents: 

USE my_lib.ALL 
To specify use of a package: 

USE my_lib.my_pack 
To specify use of all contents of a package: 

USE my_lib.my_pack.ALL 

Design_Organization.libraries 



LIBRARY ls7400; 
USE ls7400.simple_gates.ALL; 
-- 
ARCHITECTURE gate_level OF sr_latch IS 
SIGNAL im1, im2, im3, im4 : BIT; 
BEGIN 
g1 : n2 PORT MAP (s, c, im1); 
g2 : n2 PORT MAP (r, c , im2); 
g3 : n2 PORT MAP (im1, im4, im3); 
g4 : n2 PORT MAP (im3, im2, im4); 
q <=- im3; 
END gate_level; 



Assume library ls7400 contains simple_gates package 

Assume simple_gates package contains component declarations 

All component declarations are visible 



Design_Organization.libraries 



LIBRARY ls7400; 
USE ls7400.ALL; 
-- 
. 
. 
. . . _FOR g1, g3 : n2 
. . . _ USE ENTITY ls7400.nand2 (single_delay); 
. . . _ END FOR; 



Entities and ARCHITECTUREs in ls7400 have become visible 

nand2 (single_delay) from ls7400 is used 

In the rest of this tutorial, basic_utilities package will be used 

basic_utilities contains handy declarations and subprograms 



Type Mismatch?





No Generics Specified in Component Declarations





No Generics Specified in Component Declarations





Generic Parameters To be Passed to Lower Instances 





Higher Level Entity





ConfigurationDeclaration

